Biorthogonal filter pairs:
Bspline filter pairs and the CDF-7/9 filter pair

Bspline functions

The Bspline functions are generated by iterated convolution from the box function. Calling the procedure `bspline[N,t]` constructs the first N+1 Bspline functions as piecewise polynomials of degree k ($0 \le k \le N$)

```
bspline[N_, t_] := Module[{bsp, k, s},
  (*Bspline functions by convolution*)
  bsp[0] = UnitBox[t];
  If[N == 0, Return[{bsp[0]}]];
  Do[
   bsp[k] = Integrate[bsp[k - 1] /. (t → t - s), {s, -1 / 2, 1 / 2}],
   {k, 1, N}];
  Return[Table[bsp[k], {k, 0, N}]]
 ]
```

The first 3 Bspline functions as piecewise polynomial functions

```
bspline[2, t]
```

$$\left\{\text{UnitBox}[t],\ \begin{cases} 1 & t == 0 \\ 1-t & 0<t<1 \\ 1+t & -1<t<0 \\ 0 & \text{True}\end{cases},\ \begin{cases} \frac{1}{4}\left(3-4t^2\right) & -\frac{1}{2}<t<\frac{1}{2} \\ \frac{1}{8}\left(3+4t-4t^2\right) & t == \frac{1}{2} \\ \frac{1}{8}\left(9-12t+4t^2\right) & \frac{1}{2}<t<\frac{3}{2} \\ \frac{1}{8}\left(9+12t+4t^2\right) & -\frac{3}{2}<t\le-\frac{1}{2} \\ 0 & \text{True}\end{cases}\right\}$$

Plotting the first 3 Bspline functions

```
Plot[%, {t, -2, 2}, Filling → Axis]
```



Computing the Fourier transform of the first N+1 Bspline functions

```
bsplineft[N_, s_] := Module[{bsp, ft, t},
  (*The Fourier transforms of the Bspline functions*)
  bsp = bspline[N, t];
  ft = Map[FourierTransform[#, t, s, FourierParameters → {0, -2 Pi}] &, bsp];
  Simplify[ExpToTrig[ft]]
 ]
```

The case N=2: iterated convolution means powers for the Fourier transforms

```
bsplineft[2, s]
```

$$\left\{\mathrm{Sinc}[\pi s], \frac{\mathrm{Sin}[\pi s]^2}{\pi^2 s^2}, \frac{\mathrm{Sin}[\pi s]^3}{\pi^3 s^3}\right\}$$

```
Plot[%, {s, -5, 5}, Filling → Axis, PlotRange → All]
```



Bspline filters

Defining Bspline filters via powers of the cosine -- note the difference between the cases of even and odd N

```
H[N_, w_] := Module[{res},
   (*Frequency representation of the Bspline filters*)
   res  = Sqrt[2] Cos[w / 2] ^N;
   If[Mod[N, 2] == 1, res = Exp[I w / 2] res];
   Collect[TrigToExp[res], \:08]E
  ]
```

The first 5 Bspline filters -- defining expression and expanded form

```
Table[{N, H[N, ω], Expand[H[N, ω]]}, {N, 0, 5}] // MatrixForm
```

0	$\sqrt{2}$	$\sqrt{2}$
1	$\frac{1}{\sqrt{2}} + \frac{e^{i\omega}}{\sqrt{2}}$	$\frac{1}{\sqrt{2}} + \frac{e^{i\omega}}{\sqrt{2}}$
2	$\frac{1}{\sqrt{2}} + \frac{e^{-i\omega}}{2\sqrt{2}} + \frac{e^{i\omega}}{2\sqrt{2}}$	$\frac{1}{\sqrt{2}} + \frac{e^{-i\omega}}{2\sqrt{2}} + \frac{e^{i\omega}}{2\sqrt{2}}$
3	$\frac{e^{\frac{i\omega}{2}}\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^3}{4\sqrt{2}}$	$\frac{3}{4\sqrt{2}} + \frac{e^{-i\omega}}{4\sqrt{2}} + \frac{3\,e^{i\omega}}{4\sqrt{2}} + \frac{e^{2i\omega}}{4\sqrt{2}}$
4	$\frac{\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^4}{8\sqrt{2}}$	$\frac{3}{4\sqrt{2}} + \frac{e^{-i\omega}}{2\sqrt{2}} + \frac{e^{i\omega}}{2\sqrt{2}} + \frac{e^{-2i\omega}}{8\sqrt{2}} + \frac{e^{2i\omega}}{8\sqrt{2}}$
5	$\frac{e^{\frac{i\omega}{2}}\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^5}{16\sqrt{2}}$	$\frac{5}{8\sqrt{2}} + \frac{5\,e^{-i\omega}}{16\sqrt{2}} + \frac{5\,e^{i\omega}}{8\sqrt{2}} + \frac{e^{-2i\omega}}{16\sqrt{2}} + \frac{5\,e^{2i\omega}}{16\sqrt{2}} + \frac{e^{3i\omega}}{16\sqrt{2}}$

Plotting the frequency representation

```
Plot[{Abs[H[1, ω]], H[2, ω], Abs[H[3, ω]], H[8, ω]}, {ω, 0, Pi}]
```



Showing the sequence of filter coefficients of the Bspline filters

```
h[N_] := Module[{cl, w},
  (*Filter coefficients of the Bspline filters*)
  cl = CoefficientList[
    Expand[H[N, w] Exp[Floor[N / 2] I w]], E^(I w)];
  Table[{-Floor[N / 2] + k, cl[[k + 1]]}, {k, 0, N}]
 ]
```

```
h[2]
```

$$\left\{\left\{-1, \frac{1}{2\sqrt{2}}\right\}, \left\{0, \frac{1}{\sqrt{2}}\right\}, \left\{1, \frac{1}{2\sqrt{2}}\right\}\right\}$$

```
h[3]
```

$$\left\{\left\{-1, \frac{1}{4\sqrt{2}}\right\}, \left\{0, \frac{3}{4\sqrt{2}}\right\}, \left\{1, \frac{3}{4\sqrt{2}}\right\}, \left\{2, \frac{1}{4\sqrt{2}}\right\}\right\}$$

```
h[4]
```

$$\left\{\left\{-2, \frac{1}{8\sqrt{2}}\right\}, \left\{-1, \frac{1}{2\sqrt{2}}\right\}, \left\{0, \frac{3}{4\sqrt{2}}\right\}, \left\{1, \frac{1}{2\sqrt{2}}\right\}, \left\{2, \frac{1}{8\sqrt{2}}\right\}\right\}$$

Orthogonal partners of Bspline filters

Defining the Daubechies polynomials

```
Daub[N_, z_] := Sum[Binomial[N + k, k] z^k, {k, 0, N}]
```

```
Daub[3, z]
```

$$1 + 4 z + 10 z^2 + 20 z^3$$

Computing the frequency representation of a filter of length 2M+N-1 which forms a biorthogonal pair with the Bspline filter of length N+1 by using the construction with Daubechies polynomials

```
K[M_, N_, w_] := Module[{res, n, m},
  (* Frequency representation of biorthogogonal filter partners*)
  If[Mod[N - M, 2] != 0, Throw["falsche Parität"]];
  If[Mod[N, 2] == 0,
   n = N / 2;
   m = M / 2;
   res = Sqrt[2] Cos[w / 2] ^M Daub[n + m - 1, Sin[w / 2] ^2]];
  If[Mod[N, 2] == 1,
   n = (N - 1) / 2;
   m = (M - 1) / 2;
   res = Sqrt[2] Exp[I w / 2] Cos[w / 2] ^M Daub[n + m, Sin[w / 2] ^2]];
  Expand[TrigToExp[res]]
 ]
```

Some examples of biorthogonal partners

K[1, 1, ω]

$$\frac{1}{\sqrt{2}}+\frac{e^{i\omega}}{\sqrt{2}}$$

K[1, 3, ω]

$$-\frac{1}{2\sqrt{2}}+\sqrt{2}-\frac{e^{-i\omega}}{2\sqrt{2}}-\frac{e^{i\omega}}{2\sqrt{2}}+\sqrt{2}\,e^{i\omega}-\frac{e^{2i\omega}}{2\sqrt{2}}$$

K[2, 2, ω]

$$-\frac{1}{2\sqrt{2}}+\sqrt{2}+\frac{e^{-i\omega}}{2\sqrt{2}}+\frac{e^{i\omega}}{2\sqrt{2}}-\frac{e^{-2i\omega}}{4\sqrt{2}}-\frac{e^{2i\omega}}{4\sqrt{2}}$$

K[5, 3, ω]

$$\frac{175}{128\sqrt{2}}-\frac{13\,e^{-i\omega}}{128\sqrt{2}}+\frac{175\,e^{i\omega}}{128\sqrt{2}}-\frac{97\,e^{-2i\omega}}{256\sqrt{2}}-\frac{13\,e^{2i\omega}}{128\sqrt{2}}+\frac{19\,e^{-3i\omega}}{256\sqrt{2}}-\frac{97\,e^{3i\omega}}{256\sqrt{2}}+\frac{15\,e^{-4i\omega}}{256\sqrt{2}}+\frac{19\,e^{4i\omega}}{256\sqrt{2}}-\frac{5\,e^{-5i\omega}}{256\sqrt{2}}+\frac{15\,e^{5i\omega}}{256\sqrt{2}}-\frac{5\,e^{6i\omega}}{256\sqrt{2}}$$

K[4, 6, ω]

$$\frac{987}{256\sqrt{2}}+\frac{63\,e^{-i\omega}}{256\sqrt{2}}+\frac{63\,e^{i\omega}}{256\sqrt{2}}-\frac{1827\,e^{-2i\omega}}{1024\sqrt{2}}-\frac{1827\,e^{2i\omega}}{1024\sqrt{2}}+\frac{235\,e^{-3i\omega}}{512\sqrt{2}}+\frac{235\,e^{3i\omega}}{512\sqrt{2}}+\frac{165\,e^{-4i\omega}}{512\sqrt{2}}+\frac{165\,e^{4i\omega}}{512\sqrt{2}}-\frac{105\,e^{-5i\omega}}{512\sqrt{2}}-\frac{105\,e^{5i\omega}}{512\sqrt{2}}+\frac{35\,e^{-6i\omega}}{1024\sqrt{2}}+\frac{35\,e^{6i\omega}}{1024\sqrt{2}}$$

Checking the orthogonality condition

The case N=2, M=2

```
H[2, ω] Conjugate[K[2, 2, ω]] + H[2, ω + Pi] Conjugate[K[2, 2, ω + Pi]]
```

$$\left(\frac{1}{\sqrt{2}}+\frac{e^{-i(\pi+\omega)}}{2\sqrt{2}}+\frac{e^{i(\pi+\omega)}}{2\sqrt{2}}\right)\left(-\frac{1}{2\sqrt{2}}+\sqrt{2}-\frac{e^{-i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\frac{e^{i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}\right)+\left(\frac{1}{\sqrt{2}}+\frac{e^{-i\omega}}{2\sqrt{2}}+\frac{e^{i\omega}}{2\sqrt{2}}\right)\left(-\frac{1}{2\sqrt{2}}+\sqrt{2}+\frac{e^{-i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}+\frac{e^{i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

2

The case N=2, M=4

```
H[2, ω] Conjugate[K[4, 2, ω]] + H[2, ω + Pi] Conjugate[K[4, 2, ω + Pi]]
```

$$\left(\frac{1}{\sqrt{2}}+\frac{e^{-i\omega}}{2\sqrt{2}}+\frac{e^{i\omega}}{2\sqrt{2}}\right)\left(\frac{45}{32\sqrt{2}}+\frac{19\,e^{-i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}+\frac{19\,e^{i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{3\,e^{-3i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{3\,e^{3i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}+\frac{3\,e^{-4i\,\text{Conjugate}[\omega]}}{64\sqrt{2}}+\frac{3\,e^{4i\,\text{Conjugate}[\omega]}}{64\sqrt{2}}\right)+\left(\frac{1}{\sqrt{2}}+\frac{e^{-i(\pi+\omega)}}{2\sqrt{2}}+\frac{e^{i(\pi+\omega)}}{2\sqrt{2}}\right)\left(\frac{45}{32\sqrt{2}}-\frac{19\,e^{-i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{19\,e^{i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}+\frac{3\,e^{-3i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}+\frac{3\,e^{3i\,\text{Conjugate}[\omega]}}{32\sqrt{2}}+\frac{3\,e^{-4i\,\text{Conjugate}[\omega]}}{64\sqrt{2}}+\frac{3\,e^{4i\,\text{Conjugate}[\omega]}}{64\sqrt{2}}\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

2

The case N=4, M=2

```
H[4, ω] Conjugate[K[2, 4, ω]] + H[4, ω + Pi] Conjugate[K[2, 4, ω + Pi]]
```

$$\frac{1}{8\sqrt{2}}\left(e^{-\frac{1}{2}i(\pi+\omega)}+e^{\frac{1}{2}i(\pi+\omega)}\right)^4\left(\frac{5}{2\sqrt{2}}-\frac{5e^{-i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{5e^{i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{3e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{3e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{3e^{-3i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{3e^{3i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}\right)+\frac{1}{8\sqrt{2}}\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^4\left(\frac{5}{2\sqrt{2}}+\frac{5e^{-i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}+\frac{5e^{i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{3e^{-2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{3e^{2i\,\text{Conjugate}[\omega]}}{4\sqrt{2}}+\frac{3e^{-3i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}+\frac{3e^{3i\,\text{Conjugate}[\omega]}}{16\sqrt{2}}\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

2

The case N=3, M=1

```
H[3, ω] Conjugate[K[1, 3, ω]] + H[3, ω + Pi] Conjugate[K[1, 3, ω + Pi]]
```

$$\frac{1}{4\sqrt{2}}e^{\frac{i\omega}{2}}\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^3\left(-\frac{1}{2\sqrt{2}}+\sqrt{2}-\frac{e^{-i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}+\sqrt{2}\,e^{-i\,\text{Conjugate}[\omega]}-\frac{e^{i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}\right)+\left(\frac{3}{4\sqrt{2}}+\frac{e^{-i(\pi+\omega)}}{4\sqrt{2}}+\frac{3e^{i(\pi+\omega)}}{4\sqrt{2}}+\frac{e^{2i(\pi+\omega)}}{4\sqrt{2}}\right)\left(-\frac{1}{2\sqrt{2}}+\sqrt{2}+\frac{e^{-i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\sqrt{2}\,e^{-i\,\text{Conjugate}[\omega]}+\frac{e^{i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}-\frac{e^{-2i\,\text{Conjugate}[\omega]}}{2\sqrt{2}}\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

2

Filter properties of biorthogonal partners

```
k[M_, N_] := Module[{L},
  (*Filter coeffizients of biorthogonal filter partners*)
  L = 2 M + N - 1;
  cl = CoefficientList[
    Expand[Exp[Floor[(L - 1) / 2] I w] TrigToExp[K[M, N, w]]], E^(I w)];
  Table[{-Floor[(L - 1) / 2] + k - 1, cl[[k]]}, {k, 1, L}]
 ]
```

The case N=3, M=3

```
k[3, 3]
```

$$\left\{\left\{-3, \frac{3}{32\sqrt{2}}\right\}, \left\{-2, -\frac{9}{32\sqrt{2}}\right\}, \left\{-1, -\frac{7}{32\sqrt{2}}\right\}, \left\{0, \frac{45}{32\sqrt{2}}\right\}, \left\{1, \frac{45}{32\sqrt{2}}\right\}, \left\{2, -\frac{7}{32\sqrt{2}}\right\}, \left\{3, -\frac{9}{32\sqrt{2}}\right\}, \left\{4, \frac{3}{32\sqrt{2}}\right\}\right\}$$

Plotting the frequency representation: the cases N=1 and M=1,3,5,7

```
Plot[{Abs[K[1, 1, w]], Abs[K[3, 1, w]],
  Abs[K[5, 1, w]], Abs[K[7, 1, w]]}, {w, 0, Pi}]
```



Plotting the frequency representation: the cases N=2 and M=2,4,6,8

```
Plot[{Abs[K[2, 2, w]], Abs[K[4, 2, w]],
  Abs[K[6, 2, w]], Abs[K[8, 2, w]]}, {w, 0, Pi}]
```



Plotting the frequency representation: the cases N=3 and M=1,3,5,7

```
Plot[{Abs[K[1, 3, w]], Abs[K[3, 3, w]],
  Abs[K[5, 3, w]], Abs[K[7, 3, w]]}, {w, 0, Pi}]
```



Plotting the frequency representation: the cases M=4 and N=2,4,6,8

```
Plot[{Abs[K[4, 2, w]], Abs[K[4, 4, w]],
  Abs[K[4, 6, w]], Abs[K[4, 8, w]]}, {w, 0, Pi}]
```



A biorthogonal 7/9 filter pair

The Bspline filter coefficients for N=6

```
h[6]
```

$$\left\{\left\{-3, \frac{1}{32\sqrt{2}}\right\}, \left\{-2, \frac{3}{16\sqrt{2}}\right\}, \left\{-1, \frac{15}{32\sqrt{2}}\right\}, \left\{0, \frac{5}{8\sqrt{2}}\right\}, \left\{1, \frac{15}{32\sqrt{2}}\right\}, \left\{2, \frac{3}{16\sqrt{2}}\right\}, \left\{3, \frac{1}{32\sqrt{2}}\right\}\right\}$$

The filter coefficients of the biorthogonal partner with M=2, so that this filter has length 2M+N-1=9

```
k[2, 6]
```

$$\left\{\left\{-4, -\frac{5}{32\sqrt{2}}\right\}, \left\{-3, \frac{15}{16\sqrt{2}}\right\}, \left\{-2, -\frac{7}{4\sqrt{2}}\right\}, \left\{-1, -\frac{7}{16\sqrt{2}}\right\}, \left\{0, \frac{77}{16\sqrt{2}}\right\}, \left\{1, -\frac{7}{16\sqrt{2}}\right\}, \left\{2, -\frac{7}{4\sqrt{2}}\right\}, \left\{3, \frac{15}{16\sqrt{2}}\right\}, \left\{4, -\frac{5}{32\sqrt{2}}\right\}\right\}$$

Plotting the frequency representation

```
Plot[{Abs[H[6, w]], K[2, 6, w]}, {w, 0, Pi}]
```



Checking the orthogonality condition

```
H[6, ω] Conjugate[K[2, 6, ω]] + H[6, ω + Pi] Conjugate[K[2, 6, ω + Pi]]
```

$$\frac{1}{32\sqrt{2}}\left(e^{-\frac{1}{2}i(\pi+\omega)}+e^{\frac{1}{2}i(\pi+\omega)}\right)^6\left(\frac{77}{16\sqrt{2}}+\frac{7e^{-i\,\mathrm{Conjugate}[\omega]}}{16\sqrt{2}}+\frac{7e^{i\,\mathrm{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{7e^{-2i\,\mathrm{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{7e^{2i\,\mathrm{Conjugate}[\omega]}}{4\sqrt{2}}-\left(15e^{-3i\,\mathrm{Conjugate}[\omega]}\right)\Big/\left(16\sqrt{2}\right)-\left(15e^{3i\,\mathrm{Conjugate}[\omega]}\right)\Big/\left(16\sqrt{2}\right)-\frac{5e^{-4i\,\mathrm{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{5e^{4i\,\mathrm{Conjugate}[\omega]}}{32\sqrt{2}}\right)+$$
$$\frac{1}{32\sqrt{2}}\left(e^{-\frac{i\omega}{2}}+e^{\frac{i\omega}{2}}\right)^6\left(\frac{77}{16\sqrt{2}}-\frac{7e^{-i\,\mathrm{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{7e^{i\,\mathrm{Conjugate}[\omega]}}{16\sqrt{2}}-\frac{7e^{-2i\,\mathrm{Conjugate}[\omega]}}{4\sqrt{2}}-\frac{7e^{2i\,\mathrm{Conjugate}[\omega]}}{4\sqrt{2}}+\left(15e^{-3i\,\mathrm{Conjugate}[\omega]}\right)\Big/\left(16\sqrt{2}\right)+\left(15e^{3i\,\mathrm{Conjugate}[\omega]}\right)\Big/\left(16\sqrt{2}\right)-\frac{5e^{-4i\,\mathrm{Conjugate}[\omega]}}{32\sqrt{2}}-\frac{5e^{4i\,\mathrm{Conjugate}[\omega]}}{32\sqrt{2}}\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

2

Using the cascade algorithm (*n* iterations, discrete version) to plot the scaling function belonging to the filter *h*

```
cascadephidis[h_, n_] :=
 Module[{degrees, min, max, sum, hmodif, hpol, hpols, tbl, clist, z},
  (*Approximation of the scaling function belonging
   to the filter h by using the cascade algorithm*)
  degrees = Map[First[#] &, h];
  min = Min[degrees];
  max = Max[degrees];
  sum = Last[Total[h]];
  hmodif = Map[{First[#], Last[#] / sum} &, h];
  hpol := Apply[Plus, Map[Last[#] z^First[#] &, hmodif]];
  hpols := Expand[Product[hpol /. z → z^(2^t), {t, 0, n - 1}]];
  clist = 2^n CoefficientList[z^(-min (2^n - 1)) hpols, z];
  tbl = Table[{(k - 1) 2^(-n) , clist[[k]]}, {k, 1, Length[clist]}];
  ListPlot[tbl, PlotRange → All, Filling → Axis]
 ]
```

Using the cascade algorithm (*n* iterations, discrete version) to plot the wavelet function belonging to the filter *h*

```
cascadepsidis[h_, n_] :=
 Module[{degrees, min, max, sum, hmodif, gpol, hpol, hpols, clist, tbl, z},
  (*Approximation of the wavelet function belonging
   to the filter h by using the cascade algorithm*)
  degrees = Map[First[#] &, h];
  min = Min[degrees];
  max = Max[degrees];
  sum = Last[Total[h]];
  hmodif = Map[{First[#], Last[#] / sum} &, h];
  hpol := Apply[Plus, Map[Last[#] z^First[#] &, hmodif]];
  gpol := hpol /. (z → -1 / z);
  hpols :=
   Expand[Product[hpol /. z → z^(2^t), {t, 0, n - 2}] (gpol /. z → z^(2^(n - 1)))];
  clist = 2^n CoefficientList[z^(-min (2^(n - 1) - 1) + 2^(n - 1) max) hpols, z];
  tbl = Table[{ (k - 1) 2^(-n) , clist[[k]]}, {k, 1, Length[clist]}];
  ListPlot[tbl, PlotRange → All, Filling → Axis]
 ]
```

The scaling function belonging to the Bspline filter for N=6

```
cascadephidis[h[6], 6]
```



The wavelet function belonging to the Bspline filter for N=6

```
cascadepsidis[h[6], 4]
```



The scaling function belonging to the biorthogonal Bspline filter partner for N=6 and M=2

```
cascadephidis[k[2, 6], 3]
```



The wavelet function belonging to the biorthogonal Bspline partner filter for N=6 and M=2

```
cascadepsidis[k[2, 6], 4]
```



The CDF-7/9 filter pair

Construction

Starting with the Daubechies polynomial

```
Daub[3, z]
```

$1 + 4\,z + 10\,z^2 + 20\,z^3$

Determining the complex roots of this polynomial

```
roots = NSolve[Daub[3, z] == 0, z]
```

{{z → -0.342384}, {z → -0.078808 - 0.373931 i}, {z → -0.078808 + 0.373931 i}}

There is one real root and two complex conjugate roots. So the Daubechies polynomial can be factored into two polynomials with real coefficients: a linear polynomial *p(z)* and a quadratic polynomial *q(z).* The parameter *a* (which is the leading coefficient of the polynomial *p(z)*) will be fixed later.

```
p[z_] = a (z - (z /. roots[[1]]))
```

a (0.342384 + z)

```
q[z_] = (20 / a) (z - (z /. roots[[2]])) (z - (z /. roots[[3]]))
```

$\frac{1}{a} 20\,((0.078808 - 0.373931\,i) + z)\,((0.078808 + 0.373931\,i) + z)$

```
Expand[%]
```

$\frac{1}{a}\left(2.9207 - 1.11022 \times 10^{-16}\,i\right) + \frac{(3.15232 + 0.\,i)\,z}{a} + \frac{20\,z^2}{a}$

```
Chop[%]
```

$\frac{2.9207}{a} + \frac{3.15232\,z}{a} + \frac{20\,z^2}{a}$

The frequency representation of the H-filter that goes with the factor *p(z)*

```
H[w_] = Sqrt[2] Cos[w / 2] ^ 4 p[Sin[w / 2] ^ 2]
```

$\sqrt{2}\;a\,\mathrm{Cos}\left[\frac{w}{2}\right]^4 \left(0.342384 + \mathrm{Sin}\left[\frac{w}{2}\right]^2\right)$

The frequency representation of the K-filter that goes with the factor *q(z)*

```
K[w_] = Chop[Expand[Sqrt[2] Cos[w / 2] ^4 q[Sin[w / 2] ^2]]]
```

$$\frac{4.13049\,\text{Cos}\left[\frac{w}{2}\right]^4}{a}+\frac{1}{a}4.45805\,\text{Cos}\left[\frac{w}{2}\right]^4\text{Sin}\left[\frac{w}{2}\right]^2+\frac{1}{a}20\sqrt{2}\,\text{Cos}\left[\frac{w}{2}\right]^4\text{Sin}\left[\frac{w}{2}\right]^4$$

Evaluating *H(0)* (which should be $\sqrt{2}$)

```
H[0]
```

0.484204 a

```
const = Solve[{H[0] == Sqrt[2]}, {a}]
```

$\{\{a\to 2.9207\}\}$

The Fourier series if the H-filter after fixing the constant

```
H[w_] = H[w] /. const[[1]]
```

$$4.13049\,\text{Cos}\left[\frac{w}{2}\right]^4\left(0.342384+\text{Sin}\left[\frac{w}{2}\right]^2\right)$$

The Fourier series of the K-filter after fixing the constant

```
K[w_] = K[w] /. const[[1]]
```

$$1.41421\,\text{Cos}\left[\frac{w}{2}\right]^4+1.52637\,\text{Cos}\left[\frac{w}{2}\right]^4\text{Sin}\left[\frac{w}{2}\right]^2+9.68408\,\text{Cos}\left[\frac{w}{2}\right]^4\text{Sin}\left[\frac{w}{2}\right]^4$$

Checking that all is correct

```
K[0]
```

1.41421

Checking orthogonality conditions for the CDF-7/9 filter pair

```
H[ω] Conjugate[K[ω]] + H[ω + Pi] Conjugate[K[ω + Pi]]
```

$$4.13049\,\text{Cos}\left[\frac{\omega}{2}\right]^4\left(\text{Conjugate}\left[1.52637\,\text{Cos}\left[\frac{\omega}{2}\right]^4\text{Sin}\left[\frac{\omega}{2}\right]^2+9.68408\,\text{Cos}\left[\frac{\omega}{2}\right]^4\text{Sin}\left[\frac{\omega}{2}\right]^4\right]+1.41421\,\text{Cos}\left[\frac{1}{2}\text{Conjugate}[\omega]\right]^4\right)\left(0.342384+\text{Sin}\left[\frac{\omega}{2}\right]^2\right)+4.13049\,\text{Cos}\left[\frac{\pi+\omega}{2}\right]^4\left(\text{Conjugate}\left[1.52637\,\text{Cos}\left[\frac{\pi+\omega}{2}\right]^4\text{Sin}\left[\frac{\pi+\omega}{2}\right]^2+9.68408\,\text{Cos}\left[\frac{\pi+\omega}{2}\right]^4\text{Sin}\left[\frac{\pi+\omega}{2}\right]^4\right]+1.41421\,\text{Cos}\left[\frac{1}{2}(\pi+\text{Conjugate}[\omega])\right]^4\right)\left(0.342384+\text{Sin}\left[\frac{\pi+\omega}{2}\right]^2\right)$$

```
Assuming[ω ∈ Reals, Simplify[%]]
```

$$4.13049\,\text{Cos}\left[\frac{\omega}{2}\right]^4\left(0.342384+\text{Sin}\left[\frac{\omega}{2}\right]^2\right)\left(1.41421\,\text{Cos}\left[\frac{\omega}{2}\right]^4+\left(0.605255+0.0953979\,\text{Csc}\left[\frac{\omega}{2}\right]^2\right)\text{Sin}[\omega]^4\right)+4.13049\left(0.342384+\text{Cos}\left[\frac{\omega}{2}\right]^2\right)\text{Sin}\left[\frac{\omega}{2}\right]^4\left(1.41421\,\text{Sin}\left[\frac{\omega}{2}\right]^4+\left(0.605255+0.0953979\,\text{Sec}\left[\frac{\omega}{2}\right]^2\right)\text{Sin}[\omega]^4\right)$$

```
FullSimplify[Expand[%]]
```

$2. + 3.33067 \times 10^{-16} \text{Cos}[2\,\omega] + 2.77556 \times 10^{-17} \text{Cos}[3\,\omega] - 2.08167 \times 10^{-17} \text{Cos}[4\,\omega] - 3.46945 \times 10^{-18} \text{Cos}[5\,\omega]$

```
Chop[%]
```

2.

Properties of the CDF-7/9 filter pair

```
TrigToExp[H[ω]]
```

$0.788486 + 0.418092\, e^{-i\omega} + 0.418092\, e^{i\omega} - 0.0406894\, e^{-2 i\omega} - 0.0406894\, e^{2 i\omega} - 0.0645389\, e^{-3 i\omega} - 0.0645389\, e^{3 i\omega}$

```
hcoeffs = CoefficientList[Expand[Exp[3 I ω] %], E^{I ω}]
```

{-0.0645389, -0.0406894, 0.418092, 0.788486, 0.418092, -0.0406894, -0.0645389}

The filter coefficients of the H-filter of length 7

```
hfilter = Table[{k - 4, hcoeffs[[k]]}, {k, 1, 7}]
```

{{-3, -0.0645389}, {-2, -0.0406894}, {-1, 0.418092},
{0, 0.788486}, {1, 0.418092}, {2, -0.0406894}, {3, -0.0645389}}

```
TrigToExp[K[ω]]
```

$0.852699 + 0.377403\, e^{-i\omega} + 0.377403\, e^{i\omega} - 0.110624\, e^{-2 i\omega} - 0.110624\, e^{2 i\omega} - 0.0238495\, e^{-3 i\omega} - 0.0238495\, e^{3 i\omega} + 0.0378285\, e^{-4 i\omega} + 0.0378285\, e^{4 i\omega}$

```
kcoeffs = Re[CoefficientList[Expand[Exp[4 I ω] %], E^{I ω}]]
```

{0.0378285, -0.0238495, -0.110624, 0.377403,
0.852699, 0.377403, -0.110624, -0.0238495, 0.0378285}

The filter coefficients of the K-filter of length 7

```
kfilter = Table[{k - 5, kcoeffs[[k]]}, {k, 1, 9}]
```

```
{{-4, 0.0378285}, {-3, -0.0238495}, {-2, -0.110624}, {-1, 0.377403},
 {0, 0.852699}, {1, 0.377403}, {2, -0.110624}, {3, -0.0238495}, {4, 0.0378285}}
```

```
Plot[{H[ω], K[ω]}, {ω, 0, Pi}]
```



The scaling function belonging to the H-filter of length 7

```
cascadephidis[hfilter, 5]
```



The wavelet function belonging to the H-filter of length 7

```
cascadepsidis[hfilter, 5]
```



The scaling function belonging to the K-filter of length 9

```
cascadephidis[kfilter, 5]
```



The wavelet function belonging to the K-filter of length 9

```
cascadepsidis[kfilter, 5]
```



Low-pass conditions for the H-filter

```
{H'[ω], H'[ω] /. ω → Pi}
```

$$\left\{4.13049 \operatorname{Cos}\left[\frac{\omega}{2}\right]^5 \operatorname{Sin}\left[\frac{\omega}{2}\right] - 8.26098 \operatorname{Cos}\left[\frac{\omega}{2}\right]^3 \operatorname{Sin}\left[\frac{\omega}{2}\right] \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right),\ 0.\right\}$$

```
{H''[ω], H''[ω] /. ω → Pi}
```

$$\left\{2.06524 \operatorname{Cos}\left[\frac{\omega}{2}\right]^6 - 18.5872 \operatorname{Cos}\left[\frac{\omega}{2}\right]^4 \operatorname{Sin}\left[\frac{\omega}{2}\right]^2 - 4.13049 \operatorname{Cos}\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right) + 12.3915 \operatorname{Cos}\left[\frac{\omega}{2}\right]^2 \operatorname{Sin}\left[\frac{\omega}{2}\right]^2 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right),\ 0.\right\}$$

```
{H'''[ω], H'''[ω] /. ω → Pi}
```

$$\left\{-28.9134 \operatorname{Cos}\left[\frac{\omega}{2}\right]^5 \operatorname{Sin}\left[\frac{\omega}{2}\right] + 49.5659 \operatorname{Cos}\left[\frac{\omega}{2}\right]^3 \operatorname{Sin}\left[\frac{\omega}{2}\right]^3 + 20.6524 \operatorname{Cos}\left[\frac{\omega}{2}\right]^3 \operatorname{Sin}\left[\frac{\omega}{2}\right] \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right) - 12.3915 \operatorname{Cos}\left[\frac{\omega}{2}\right] \operatorname{Sin}\left[\frac{\omega}{2}\right]^3 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right),\ 0.\right\}$$

```
{H''''[ω], H''''[ω] /. ω → Pi}
```

$$\left\{-14.4567 \operatorname{Cos}\left[\frac{\omega}{2}\right]^6 + 167.285 \operatorname{Cos}\left[\frac{\omega}{2}\right]^4 \operatorname{Sin}\left[\frac{\omega}{2}\right]^2 - 86.7403 \operatorname{Cos}\left[\frac{\omega}{2}\right]^2 \operatorname{Sin}\left[\frac{\omega}{2}\right]^4 + 10.3262 \operatorname{Cos}\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right) - 49.5659 \operatorname{Cos}\left[\frac{\omega}{2}\right]^2 \operatorname{Sin}\left[\frac{\omega}{2}\right]^2 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right) + 6.19573 \operatorname{Sin}\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \operatorname{Sin}\left[\frac{\omega}{2}\right]^2\right),\ 8.31705\right\}$$

Low-pass conditions for the K-filter

{K'[ω], K'[ω] /. ω → Pi}

$$\left\{-2.82843\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right] + 1.52637\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^5 \mathrm{Sin}\left[\frac{\omega}{2}\right] - 3.05273\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right]^3 + 19.3682\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^5 \mathrm{Sin}\left[\frac{\omega}{2}\right]^3 - 19.3682\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right]^5,\ 0.\right\}$$

{K''[ω], K''[ω] /. ω → Pi}

$$\left\{-1.41421\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 + 0.763183\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^6 + 4.24264\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 - 8.39501\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 + 29.0523\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^6 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 + 4.5791\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^4 - 96.8408\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 \mathrm{Sin}\left[\frac{\omega}{2}\right]^4 + 29.0523\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^6,\ 0.\right\}$$

{K'''[ω], K'''[ω] /. ω → Pi}

$$\left\{7.07107\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right] - 10.6846\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^5 \mathrm{Sin}\left[\frac{\omega}{2}\right] + 29.0523\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^7 \mathrm{Sin}\left[\frac{\omega}{2}\right] - 4.24264\,\mathrm{Cos}\left[\frac{\omega}{2}\right] \mathrm{Sin}\left[\frac{\omega}{2}\right]^3 + 25.9482\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right]^3 - 280.838\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^5 \mathrm{Sin}\left[\frac{\omega}{2}\right]^3 - 4.5791\,\mathrm{Cos}\left[\frac{\omega}{2}\right] \mathrm{Sin}\left[\frac{\omega}{2}\right]^5 + 280.838\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^3 \mathrm{Sin}\left[\frac{\omega}{2}\right]^5 - 29.0523\,\mathrm{Cos}\left[\frac{\omega}{2}\right] \mathrm{Sin}\left[\frac{\omega}{2}\right]^7,\ 0.\right\}$$

{K''''[ω], K''''[ω] /. ω → Pi}

$$\left\{3.53553\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 - 5.34228\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^6 + 14.5261\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^8 - 16.9706\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 + 65.6337\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 - 522.941\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^6 \mathrm{Sin}\left[\frac{\omega}{2}\right]^2 + 2.12132\,\mathrm{Sin}\left[\frac{\omega}{2}\right]^4 - 50.3701\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^4 + 1404.19\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^4 \mathrm{Sin}\left[\frac{\omega}{2}\right]^4 + 2.28955\,\mathrm{Sin}\left[\frac{\omega}{2}\right]^6 - 522.941\,\mathrm{Cos}\left[\frac{\omega}{2}\right]^2 \mathrm{Sin}\left[\frac{\omega}{2}\right]^6 + 14.5261\,\mathrm{Sin}\left[\frac{\omega}{2}\right]^8,\ 18.937\right\}$$